
基于 CP-243-1 的 OPC 应用 

                                       

                                     河北惠仁 萧飞 2017.01 

    西门子新推出的 S7-200 SMART 自身集成了以太网口，笔者之前

曾以此为基础，编写了基于 OPC 的界面程序。但是对于自身不带以

太网口的 S7-200，如果也要有基于 OPC 的以太网应用，则只有加装

CP-243-1 的以太网通信处理器。以下为笔者以 CPU 224XP+ CP-243-1

为平台，并增加其他模块进行的项目试验记录。 

一、项目平台 

1、硬件平台 

   CPU-----214 2BD23-0XB8 

   DO-----232 0HD22-0XA0 

   AI-----231 0HC22-0XA0 

   CP-----6GK7 243-1EX01-0EX0 

   如图 1 所示。 

 
（图 1） 

2、软件平台 

   STEP 7-Micro/WIN 

（图 2） 

   S7-200 PC Access 

 
（图 3） 

二、CP 组态步骤 
1、模块组装 
    首先，将 4 个模块按照自左至右的顺序安装，依次为 CPU、DO、AI 和 CP，并将总线

连接完毕，实物连接见图 1。 
    其次，给 CPU 加载 AC 220V 的交流电，右边 3 个模块加载 DC 24V。 
2、软件组态 
   打开 STEP 7-Micro/WIN 软件，在“工具”菜单中找到“以太网向导”，注意不是“因特

网”。按照提示进行下一步，如图 4 所示，其中需要将 CP-243-1 模块的位置正确标识出来模

块位置标识有两种办法，一是通过“读取模块”按钮进行读取位置，二是通过模块占位进行

标识。模块占位规则：CPU 不在模块位置计算之列，CPU 右侧第一个模块位置规定为 0 号

位，向右依次位号为 1、2、3 等等，本案例中 CP-243-1 模块的位号为 2。 

 
（图 4） 

    继续“下一步”，如图 5 所示，根据实际模块的订货号选择型号。本案例的模块型号为：

6GK7 243-1EX01-0EX0。 

 
（图 5） 

    继续“下一步”，如图 6 所示，设置 CP-243-1 模块的 IP 地址为 192.168.2.10，子网掩码

255.255.255.0，模块连接类型选择为“自动检测通信”。 

（图 6） 

    继续“下一步”，如图 7 所示。本案例需要规定模块的连接数目为“1”。但是最多不能

超过 8 个连接。 

 
（图 7） 

    继续“下一步”，如图 8 所示。对上一步选择的 1 个连接进行配置，本案例选择“服务

器连接”，因为 PLC 作为服务器，上位机作为客户机；服务器或者本地的 TASP 为 10.02（其

中 02 为模块的槽位号），客户机或者远程的 TASP 为 10.00。这里的 TASP 其实对应以太网

协议中的端口号。虽然 S7-200 名义上支持以太网，但实质上底层的协议仍是 S7 协议，为了

区别这一点，此处用 TASP 和端口相区别。因为本案例 PC 机的 IP地址固定位为 192.168.2.20，
故可设为仅接受本台 PC 的连接请求。 

（图 8） 
    点击“确认”按钮，即可完成 CP-243-1 的配置，然后编译并保存。最后通过 USB/PPI
电缆将组态文件下载，之后再将 PG/PC 界面修改成以太网链路，即可使用此链路进行上传

和下载程序。 

三、调试过程 
1、PLC 侧的工作 
    将应用程序编制完毕，编译无误并保存，然后下载到 PLC 中。 
2、PC 侧的工作 
    首先，在安装过 S7-200 PC Access 的 PC 机上打开 VB，然后再“引用”菜单上作如图 9
的选择（蓝条框中 Siemens OPC DAAutomation 2.0），并点击“确定”按钮。 
    其次，输入代码，详见附录。但在条目 ID 上需要提醒，如本案例中的一个条目如下： 
ItemId(1) = "192.168.2.10:1000:1002,M10.0,BOOL,RW" ''磁体启动， 本条目中 IP 地址是 CP 
243-1 的 IP 地址，双方的 TASP 参照图 8，本地和远程是反过来使用，此处需加注意。并指

明 M10.0 是 BOOL 型，可以进行读写，即 RW。 
     按照本案例需求，进行组件布置。待调试完毕后生成可执行文件。 
3、连接调试 
使用网线（直连线即可）将 PC 和 PLC 的 CP243-1 连接起来，然后双击可执行文件。 

单击“OPC”连接按钮，会出现如图 10 的界面，一旦界面右侧的文本框中有数据显示，即

表明双方连接上并通讯正常，即可进行下一步的操作。 

 

（图 9） 

 

（图 10） 

四、附录 

Option Base 1 
Option Explicit 
 Dim WithEvents ServerObj As OPCServer  ''OPC Server 对象，连接 OPC 服务器 
 Dim GroupsObj As OPCGroups             ''OPC Groups 对象，添加 OPC 组 
 Dim GroupObj As OPCGroup              ''OPC Group 对象 
 Dim ItemsObj As OPCItems              ''OPC Item 集合 
 Dim ServerHandles() As Long           ''服务器端 OPC Item 的句柄 
 Dim ClientHandles(14) As Long         ''客户端 OPC Item 的句柄 
 Dim ItemId(14) As String              ''服务器条目 
 Dim Errors() As Long                  ''系统错误 
 
Private Sub Command1_Click() ''OPC 连接 
 If ServerObj Is Nothing Then Set ServerObj = New OPCServer        ''生成 OPC 服务器 
  If ServerObj.ServerState = OPCDisconnected Then 
  ServerObj.Connect ("S7200.OPCServer")                            ''连接服务器 
   End If 
 If GroupsObj Is Nothing Then Set GroupsObj = ServerObj.OPCGroups  ''生成组集合 
 If GroupObj Is Nothing Then Set GroupObj = GroupsObj.Add          ''生成组对象 
 If ItemsObj Is Nothing Then Set ItemsObj = GroupObj.OPCItems    ''生成组对象的条目集合 
 GroupObj.IsActive = True                                       ''设置组为活动状态 
 GroupObj.UpdateRate = 1000                            ''设置组更新周期为 1000 毫秒 
  
 ItemId(1) = "192.168.2.10:1000:1002,M10.0,BOOL,RW" ''磁体启动 
 ItemId(2) = "192.168.2.10:1000:1002,M10.1,BOOL,RW" ''磁体停止 
 ItemId(3) = "192.168.2.10:1000:1002,M10.3,BOOL,RW" ''系统复位 
 ItemId(4) = "192.168.2.10:1000:1002,M2.0,BOOL,RW" ''温度 
 ItemId(5) = "192.168.2.10:1000:1002,M2.1,BOOL,RW" ''水流 
 ItemId(6) = "192.168.2.10:1000:1002,M2.2,BOOL,RW" ''APB 
 ItemId(7) = "192.168.2.10:1000:1002,M2.3,BOOL,RW" ''缺相 
 ItemId(8) = "192.168.2.10:1000:1002,M2.4,BOOL,RW" ''负载 
 ItemId(9) = "192.168.2.10:1000:1002,M2.5,BOOL,RW" ''漏流 
 ItemId(10) = "192.168.2.10:1000:1002,VD16,REAL,RW" ''电流反馈 
 ItemId(11) = "192.168.2.10:1000:1002,VD24,REAL,RW" ''电压反馈 
 ItemId(12) = "192.168.2.10:1000:1002,VD64,REAL,RW" ''电流设定 
 ItemId(13) = "192.168.2.10:1000:1002,M14.2,BOOL,RW" ''过压 
 ItemId(14) = "192.168.2.10:1000:1002,M14.3,BOOL,RW" ''过流 
  
 ClientHandles(1) = 1 ''磁体启动句柄 
 ClientHandles(2) = 2 ''磁体停止句柄 
 ClientHandles(3) = 3 ''磁体复位句柄 
 ClientHandles(4) = 4 ''磁体温度句柄 
 ClientHandles(5) = 5 ''磁体水流句柄 
 ClientHandles(6) = 6 ''磁体 APB 句柄 

 ClientHandles(7) = 7 ''磁体缺相句柄 
 ClientHandles(8) = 8 ''磁体负载句柄 
 ClientHandles(9) = 9 ''磁体漏流句柄 
 ClientHandles(10) = 10 ''磁体电流反馈句柄 
 ClientHandles(11) = 11 ''磁体电压反馈句柄 
 ClientHandles(12) = 12 ''磁体电流设定句柄 
 ClientHandles(13) = 13 ''磁体过压句柄 
 ClientHandles(14) = 14 ''磁体过流句柄 
 
Call ItemsObj.AddItems(14, ItemId, ClientHandles, ServerHandles, Errors) ''增加 14 个条目 
  
 Timer1.Enabled = True ''使能定时器 1 
 Timer2.Enabled = True ''使能定时器 2 
  
End Sub 
 
Private Sub Command2_Click() ''启动 
  Dim TempServerHandles(1) As Long 
  Dim VValue(1) As Variant 
  Dim ErrorNr() As Long 
 If ServerObj Is Nothing Then ''如果尚未建立服务器连接，跳出函数 
  Exit Sub 
  End If 
 TempServerHandles(1) = ServerHandles(1) ''本服务器句柄等于系统服务器 1 号句柄 
 VValue(1) = 1                           ''将"1"写入 ClientHandle 为 1 的 OPC Item 
 GroupObj.SyncWrite 1, TempServerHandles, VValue, ErrorNr ''同步写 1 个数据 
End Sub 
 
Private Sub Command3_Click() ''停止 
  Dim TempServerHandles(1) As Long 
  Dim VValue(1) As Variant 
  Dim ErrorNr() As Long 
 If ServerObj Is Nothing Then ''如果尚未建立服务器连接，跳出函数 
  Exit Sub 
  End If 
 TempServerHandles(1) = ServerHandles(2) ''本服务器句柄等于系统服务器 1 号句柄 
 VValue(1) = 1                           ''将"1"写入 ClientHandle 为 1 的 OPC Item 
 GroupObj.SyncWrite 1, TempServerHandles, VValue, ErrorNr ''同步写 1 个数据 
End Sub 
 
Private Sub Command4_Click() ''OPC 断开 
  If ServerObj Is Nothing Then ''如果尚未建立服务器连接，跳出函数 
  Exit Sub 
  End If 

 Timer1.Enabled = False ''禁止定时器 2 
 Timer2.Enabled = False ''禁止定时器 2 
 GroupsObj.RemoveAll 
 ServerObj.Disconnect 
 Set ItemsObj = Nothing 
 Set GroupObj = Nothing 
 Set GroupsObj = Nothing 
 Unload Form1              ''卸载窗体 
End Sub 
 
Private Sub Command5_Click() ''设定生效 
 Dim TempServerHandles(1) As Long 
 Dim VValue(1) As Variant 
 Dim ErrorNr() As Long 
 If ServerObj Is Nothing Then ''如果尚未建立服务器连接，跳出函数 
  Exit Sub 
  End If 
 TempServerHandles(1) = ServerHandles(12) ''本服务器句柄等于系统服务器 3 号句柄 
 VValue(1) = Text5.Text                  ''将设定值写入 ClientHandle 为 3 的 OPC Item 
 GroupObj.SyncWrite 1, TempServerHandles, VValue, ErrorNr ''同步写 1 个数据 
End Sub 
 
Private Sub Command6_Click() ''复位 
    Dim TempServerHandles(1) As Long 
 Dim VValue(1) As Variant 
 Dim ErrorNr() As Long 
 If ServerObj Is Nothing Then ''如果尚未建立服务器连接，跳出函数 
  Exit Sub 
  End If 
 TempServerHandles(1) = ServerHandles(3) ''本服务器句柄等于系统服务器 4 号句柄 
 VValue(1) = 1                            ''将"1"写入 ClientHandle 为 4 的 OPC Item 
 GroupObj.SyncWrite 1, TempServerHandles, VValue, ErrorNr  ''同步写 1 个数据 
End Sub 
 
Private Sub Timer1_Timer() 
  Dim TempServerHandles(10) As Long 
  Dim Values() As Variant 
  Dim Errors() As Long 
  Dim Source As Integer 
  Source = OPCCache 
  If GroupObj Is Nothing Then 
  Exit Sub 
  End If 

  TempServerHandles(1) = ServerHandles(4) ''本服务器句柄等于系统服务器 4 号句柄 
  TempServerHandles(2) = ServerHandles(5) ''本服务器句柄等于系统服务器 5 号句柄 
  TempServerHandles(3) = ServerHandles(6) ''本服务器句柄等于系统服务器 7 号句柄 
  TempServerHandles(4) = ServerHandles(7) ''本服务器句柄等于系统服务器 8 号句柄 
  TempServerHandles(5) = ServerHandles(8) ''本服务器句柄等于系统服务器 9 号句柄 
  TempServerHandles(6) = ServerHandles(9) ''本服务器句柄等于系统服务器 10 号句柄 
  TempServerHandles(7) = ServerHandles(10) ''本服务器句柄等于系统服务器 11 号句柄 
  TempServerHandles(8) = ServerHandles(11) ''本服务器句柄等于系统服务器 12 号句柄 
  TempServerHandles(9) = ServerHandles(13) ''本服务器句柄等于系统服务器 12 号句柄 
  TempServerHandles(10) = ServerHandles(14) ''本服务器句柄等于系统服务器 12 号句柄 
   
  If GroupObj Is Nothing Then 
  Exit Sub 
  End If 
  If GroupObj.OPCItems.Count > 0 Then 
  GroupObj.SyncRead Source, 10, TempServerHandles, Values, Errors ''同步读 10 个数据 
  End If 
   
  If Values(1) Then 
  Label6.Caption = "温度故障" 
  Else 
  Label6.Caption = "温度正常" 
  End If 
  If Values(2) Then 
  Label7.Caption = "水流故障" 
  Else 
  Label7.Caption = "水流正常" 
  End If 
  If Values(3) Then 
  Label8.Caption = "APB 故障" 
  Else 
  Label8.Caption = "APB 正常" 
  End If 
  If Values(4) Then 
  Label9.Caption = "缺相" 
  Else 
   Label9.Caption = "不缺相" 
  End If 
  If Values(5) Then 
  Label10.Caption = "负载故障" 
  Else 
  Label10.Caption = "负载正常" 
  End If 
  If Values(6) Then 



13 
 

  Label11.Caption = "漏流" 
  Else 
  Label11.Caption = "不漏流" 
  End If 
   
  Text2.Text = Values(7) 
  Text1.Text = Values(8) 
   
  If Values(9) Then 
  Label3.Caption = "过压" 
  Else 
  Label3.Caption = "不过压" 
  End If 
   
  If Values(10) Then 
  Label4.Caption = "过流" 
  Else 
  Label4.Caption = "不过流" 
  End If 
   
End Sub 
 
Private Sub Timer2_Timer() 
 Dim TempServerHandles(3) As Long 
 Dim VValue(3) As Variant 
 Dim ErrorNr() As Long 
 TempServerHandles(1) = ServerHandles(1) ''本服务器句柄等于系统服务器 1 号句柄 
 TempServerHandles(2) = ServerHandles(2) ''本服务器句柄等于系统服务器 1 号句柄 
 TempServerHandles(3) = ServerHandles(3) ''本服务器句柄等于系统服务器 1 号句柄 
  
 VValue(1) = 0                            ''将"0"写入 ClientHandle 为 1 的 OPC Item 
 VValue(2) = 0 
 VValue(3) = 0 
 
 GroupObj.SyncWrite 3, TempServerHandles, VValue, ErrorNr  ''同步写 3 个数据 
End Sub 

